

Naval Research Laboratory

Stennis Space Center, MS 39529-5004



NRL/MR/7330--00-8245

Boundary Conditions in the Pacific West Coast Princeton Ocean Model of CoBALT

PETER A. ROCHFORD

*Ocean Sciences Branch
Oceanography Division*

IGOR SHULMAN

*University of Southern Mississippi
Stennis Space Center, Mississippi*

July 19, 2000

20000727 211

Approved for public release; distribution is unlimited.

REPORT DOCUMENTATION PAGE

*Form Approved
OMB No. 0704-0188*

Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project (0704-0188), Washington, DC 20503.

1. AGENCY USE ONLY (Leave Blank)	2. REPORT DATE	3. REPORT TYPE AND DATES COVERED	
	July 19, 2000	Final	
4. TITLE AND SUBTITLE Boundary Conditions in the Pacific West Coast Princeton Ocean Model of CoBALT			5. FUNDING NUMBERS
6. AUTHOR(S) Peter A. Rochford and Igor Shulman*			
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) Naval Research Laboratory Stennis Space Center, MS 39529-5004			8. PERFORMING ORGANIZATION REPORT NUMBER NRL/MR/7330--00-8245
9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)			10. SPONSORING/MONITORING AGENCY REPORT NUMBER
11. SUPPLEMENTARY NOTES *University of Southern Mississippi Stennis Space Center, Mississippi			
12a. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release; distribution is unlimited.			12b. DISTRIBUTION CODE
13. ABSTRACT (Maximum 200 words) The boundary conditions (BCs) implemented in the Pacific West Coast (PWC) version of the Princeton Ocean Model (POM) as developed at the Naval Research Laboratory (NRL) are described. BCs for the barotropic velocities along the open boundaries are a Flather form for the normal component and an advective form for the tangential component with matching to externally supplied transports. Baroclinic velocities use a radiational BC for the normal component, and an advectional BC for the tangential component. Sea surface height, temperature, and salinity use radiational or advective BCs. Simple relaxation techniques are used for the assimilation of sea surface temperature (SST) and sea surface salinity (SSS). The interior state of the ocean is maintained by relaxation to a monthly climatology. A zero gradient boundary condition is applied for the vertical velocity, a clamped condition is applied to the turbulent kinetic energy and turbulent length scale. The explicit treatment of these BCs is given here to serve as a reference for PWC POM users.			
14. SUBJECT TERMS Boundary conditions Ocean general circulation model			15. NUMBER OF PAGES 18
			16. PRICE CODE
17. SECURITY CLASSIFICATION OF REPORT UNCLASSIFIED	18. SECURITY CLASSIFICATION OF THIS PAGE UNCLASSIFIED	19. SECURITY CLASSIFICATION OF ABSTRACT UNCLASSIFIED	20. LIMITATION OF ABSTRACT UL

CONTENTS

1. INTRODUCTION	1
2. ELEVATION	2
3. BAROTROPIC VELOCITIES	3
4. BAROCLINIC VELOCITIES	5
5. TEMPERATURE AND SALINITY	7
5.1 Water Column	7
5.2 Sea Surface Temperature and Salinity	8
5.3 Interior Relaxation to Climatology	8
6. VERTICAL VELOCITY	9
7. TURBULENT KINETIC ENERGY AND LENGTH SCALE	9
8. SUMMARY	10
9. ACKNOWLEDGEMENTS	11
10. REFERENCES	11
APPENDIX A – FORTRAN Symbols	13
APPENDIX B – Princeton Ocean Model C-grid	15

Boundary Conditions in the Pacific West Coast Princeton Ocean Model of CoBALT

1. INTRODUCTION

An ocean general circulation model (OGCM) for the Pacific West Coast (PWC) was developed by NRL in an effort to meet Navy operational needs for this region [Clancy *et al.*, 1996]. The OGCM was based on the POM with extensive modifications to include one-way coupling from the NRL Layered Ocean Model (NLOM) [Ko, 1996a; Ko *et al.*, 1996b; Ko *et al.*, 1996c; Metzger *et al.*, 1998]. The PWC POM as originally developed by Ko demonstrated sufficient predictive skill that it was delivered to Fleet Numerical Meteorology and Oceanography Center for evaluation as an operational product for use by the fleet. Since then this version has been coupled to a regional weather forecast model [Allard *et al.*, 1996], been turned into a nowcast/forecast model with data assimilation [Ko *et al.*, 1999a], and coupled to an OGCM of the North Pacific Ocean [Ko *et al.*, 1999b]. A close variant of the PWC POM is presently being used for studies as part of the Coupled Biophysical-Dynamics Across the Littoral Transition (CoBALT) project at NRL [Haidvogel *et al.*, 2000; Kindle *et al.*, 1999; Righi *et al.*, 1999; Shulman *et al.*, 1999a]. The CoBALT version of the PWC POM differs from the original due to modifications made to the BCs for the external mode as devised by Shulman *et al.* [1999b]. In late 1999 the real-time simulation at FNMOC was modified to be consistent with the CoBALT version of the PWC POM. Given the wide applicability of this OGCM, and the uniqueness of the boundary conditions that embed the coastal model within the global NLOM, we document in this report the details of the implementation of the boundary conditions for the benefit of the user community. The BCs described are those specifically used in the PWC POM of the CoBALT project, which we will hereinafter referred to as just the PWC POM.

The open boundaries along the south, north, and western edges of the PWC POM domain are forced by fields provided from the global NRL Layered Ocean Model (NLOM). The PWC POM is bounded on the eastern edge of the model domain by the Pacific west coast of the United States and includes temperature and salinity forcing as dictated by river inflows. The model also assimilates sea surface temperature (SST) from daily predictions of Multi-Channel Sea Surface Temperature (MCSST) and sea surface salinity (SSS) from the monthly climatology of the World Ocean Atlas 1994 [Levitus *et al.*, 1994; Levitus and Boyer, 1994] (hereinafter referred to as Levitus data). The state of the interior ocean is maintained via a relaxation to monthly Levitus temperature and salinity.

Section 2 presents the BCs as implemented for elevation. Sections 3 and 4 describe the BCs for the barotropic and baroclinic velocities, respectively. Section 5 explains the temperature and salinity BCs, the assimilation of MCSST and Levitus SSS, and the relaxation of the interior temperature and salinity to climatology. Section 6 presents the BCs for the vertical velocity. Section 7

gives the BCs for the turbulent kinetic energy and the turbulence length scale. Finally, in section 8 we provide a short summary of the BCs.

2. ELEVATION

For the elevation η a Sommerfeld radiation BC is applied

$$\frac{\partial \eta}{\partial t} + C_{bt} \frac{\partial \eta}{\partial n} = 0, \quad (1)$$

where n is the unit outward normal to the boundary, and C_{bt} is the barotropic phase speed. For the PWC region $n = -y$ for the southern boundary, $n = y$ for the northern boundary, and $n = -x$ for the western boundary. The phase speed is given by

$$C_{bt} = (gH)^{1/2}, \quad (2)$$

where g is the gravitational acceleration and H is the total depth of the ocean.

In the FORTRAN implementation of this BC (see BCOND subroutine in POM) a 1:2:1 filter is applied to the elevations at the boundary before the BC is applied in order to reduce noise

$$\bar{\eta}_x(i, j) = \frac{1}{4} (\eta(i-1, j) + 2\eta(i, j) + \eta(i+1, j)) \quad (3)$$

$$\bar{\eta}_y(i, j) = \frac{1}{4} (\eta(i, j-1) + 2\eta(i, j) + \eta(i, j+1)), \quad (4)$$

where $\bar{\eta}_x$ and $\bar{\eta}_y$ denote filtering in the zonal and meridional directions, respectively. The elevation at the boundaries is updated for the forward time step $t + \Delta t$ via

$$\bar{\eta}(t + \Delta t) = \bar{\eta}(t) - C_{bt} \frac{\Delta \bar{\eta}(t)}{\Delta n} \Delta t_E \quad (5)$$

where $\Delta \bar{\eta}$ represents the finite difference of the elevation and Δt_E is the external time step. This boundary condition is applied to the southern, northern, and western boundaries.

As an example, the FORTRAN code for the elevation BC at the southern boundary is given below.

```

DO I = 2, NBS
    CEES(I) = SQRT( (H(I,1)+H(I,2))*5*GRAV )
    &           * DTE / ( (DY(I,1)+DY(I,2))*5 )
    ELF(I,1) =   CEES(I) * (.25*EL(I-1,2)+.5*EL(I,2)+.25*EL(I+1,2))
    &           +(1-CEES(I))*(.25*EL(I-1,1)+.5*EL(I,1)+.25*EL(I+1,1))
END DO

```

The FORTRAN symbols are defined in Appendix A. Note that H and DY are averaged to collocate them with V on the C-grid (Appendix B).

The elevation at the SW and NW corners of the model grid are given special treatment, as they are averaged with their boundary neighbours.

$$\begin{aligned} \text{ELF}(1,1) &= (\text{ELF}(1,2)+\text{EL}(2,1))/2. \\ \text{ELF}(1,jm) &= (\text{ELF}(1,JMM1)+\text{EL}(2,JM))/2. \end{aligned}$$

This is done for numerical safety in the event these corner locations are accessed elsewhere in the model.

For numerical safety the elevations are also specified at the river locations along the coast. Their values are assigned to those of the adjacent interior grid point where they are computed by the POM because there is no suitable information on river elevations at ocean boundaries. This is coded into the POM as

```
DO I = 1, NRV
  ELF(IRV(I),JRV(I)) = ELF(IRV(I)-1,JRV(I))
END DO
```

where IRV and JRV are grid locations of the rivers.

3. BAROTROPIC VELOCITIES

For the barotropic velocity $\mathbf{V} = (U, V)$ different BCs are applied for the normal and tangential components. For the normal component V_n a radiation BC that was used for barotropic two-dimensional tide and storm surge modeling [Flather, 1976] is applied

$$V_n = V_n^0 + \left(\frac{g}{H} \right)^{1/2} (\eta - \eta_0), \quad (6)$$

where V_n^0 is the NLOM normal velocity at the open boundary, η is the model sea surface elevation calculated from the continuity equation, η_0 is the NLOM elevation at the open boundary. Note that Flather [1976] considered both meteorological (V_{met} and η_{met}) and tidal forcing (V_{tide} and η_{tide}) in the BC

$$V_n = V_{\text{met}} + V_{\text{tide}} - \left(\frac{g}{H} \right)^{1/2} (\eta - \eta_{\text{met}} - \eta_{\text{tide}}). \quad (7)$$

For the tangential component V_t an advective BC is used to allow one way external forcing of the PWC POM. The vertically averaged tangential component of velocity from the NLOM is advected into the model domain in the case of inflow, and the internal vertically averaged tangential component of velocity is advected to the open boundary in the case of outflow. This leads to a BC of the form

$$\frac{\partial V_t}{\partial t} + V_n \frac{\partial V_t}{\partial n} = 0, \quad (8)$$

where here

$$\frac{\partial V_t}{\partial n} = \begin{cases} (V_t - V_g)/\Delta x_n & V_n \geq 0 \\ (V_i - V_t)/\Delta x_n & V_n < 0 \end{cases} \quad (9)$$

with $V_g = V_{t0}/H\Delta x_n$ the tangential barotropic velocity at the open boundary obtained from the NLOM transport V_{t0} , V_i the tangential barotropic velocity at one grid point inside the open

boundary, and Δx_n the grid spacing in the direction normal to the boundary. The Flather BC is applied to V at the southern and northern boundaries and to U at the western boundary. The advectional BC is applied to U at the southern and northern boundaries and to V at the western boundary.

As an example, the FORTRAN code for the normal barotropic BC at the southern boundary is

```
DO I = 2, NBS
  DEPTH=( H(I,1)+H(I,2) )/2.
  DV   =( DX(I,1)+DX(I,2) )/2.
  COVRHS(I)= SQRT( GRAV/H(I,2) )*DEPTH*DV
  VAF(I,2) = (VAGS(I)*RAMP -
&           COVRHS(I)*( EL(I,2)-ELGS(I)*RAMP ) )/(DEPTH*DV)
ENDDO
```

while that for the tangential barotropic BC is

```
DO I = 2, NBS
  VMID = ( VA(I-1,2)+VA(I,2) )/2.
  DEPTH= ( H(I-1,1)+H(I,1) )/2.
  DV   = ( DY(I-1,1)+DY(I,1) )/2.
  UAG  = RAMP*UAGS(I)/(DEPTH*DV)
  AEUS(I) = DTE/(DY(I-1,1)+DY(I,1)+DY(I-1,2)+DY(I,2))*2.
  UAF(I,1)= UA(I,1) - AEUS(I) *
&           ( (VMID+ABS(VMID)) * (UA(I,1)-UAG      )
&           +(VMID-ABS(VMID)) * (UA(I,2)-UA(I,1)) )
END DO
```

Note that H in DEPTH, V in VMID, and Δx_n in DV are averaged over 2 grid points to collocate at the corresponding U and V points on the C-grid (Appendix B) the phase speed $(g/H)^{1/2}$ and normal velocity V_n , respectively. Similarly, the grid space averaging of Δy in AEUS is to properly account for the tangential velocity derivative $\partial V_t / \partial n$.

After the BCs have been applied to the barotropic velocities the values of V at the first row of the southern boundary and the values of U at the first column of the western boundary are assigned the values of their adjacent grid points for numerical safety.

```
DO I = 1, NBS
  VAF(I,1) = VAF(I,2)
END DO
DO J = 1, NBW
  UAF(1,J) = UAF(2,J)
END DO
```

River runoff is included as a BC for the barotropic velocities. Only U is specified because all the rivers flow from east to west. It is coded as

```
DO IJ = 1, NRV
  I = IRV(IJ)
  J = JRV(IJ)
```

```

DEPTH = (H(I-1,J)+ELF(I-1,J)+H(I,J)+ELF(I,J))/2.0
DV = (DY(I-1,J)+DY(I,J))/2.0
UAF(I,J) = RAMP*UARV(IJ)/(DEPTH*DV)
END DO

```

where UARV is the river transport.

4. BAROCLINIC VELOCITIES

The baroclinic velocity $\mathbf{v} = (u, v)$ has a radiational BC applied for the normal component v_n that is identical in form to that used for the elevation

$$\frac{\partial v_n}{\partial t} + C_{bc} \frac{\partial v_n}{\partial n} = 0, \quad (10)$$

but where the phase speed is assigned as a fraction of the barotropic phase speed

$$C_{bc} = 0.005 (gH)^{1/2}. \quad (11)$$

This explicit relation was determined through numerical tests to yield adequate results. For the tangential component v_t , an advectional BC is applied that is identical in form with that used for the barotropic tangential velocity

$$\frac{\partial v_t}{\partial t} + v_n \frac{\partial v_t}{\partial n} = 0. \quad (12)$$

The normal gradient of the velocity is given here by

$$\frac{\partial v_t}{\partial n} = \begin{cases} (v_t - v_g)/\Delta x_n & v_n \geq 0 \\ (v_i - v_t)/\Delta x_n & v_n < 0 \end{cases} \quad (13)$$

where v_i is the tangential barotropic velocity at one grid point inside the open boundary, and $v_g = 0$ because there are no vertical profiles of tangential velocity available at the open boundary. The radiational BC is applied to v at the southern and northern boundaries and to u at the western boundary. The advectional BC is applied to u at the southern and northern boundaries and to v at the western boundary.

The FORTRAN implementation of the BC for the normal baroclinic velocity at the southern and northern boundaries are given below as examples.

```

ck ** South
DO I = 2, NBS
  CIVS(I) = SQRT(H(I,2)*GRAV*5.E-3)*DTI/DY(I,2)
  VF(I,2,K) = CIVS(I) * (.25*V(I-1,3,K)+.5*V(I,3,K)+.25*V(I+1,3,K))
  &           +(1-CIVS(I))*(.25*V(I-1,2,K)+.5*V(I,2,K)+.25*V(I+1,2,K))
END DO

ck ** North
DO I = 2, NBN
  CIVN(I) = SQRT(H(I,JMM1)*GRAV*5.E-3)*DTI/DY(I,JMM1)
  VF(I,JM,K) = CIVN(I) * (.25*V(I-1,JMM1,K)+.5*V(I,JMM1,K))

```

```

&           +.25*V(I+1,JMM1,K))
&           +(1-CIVN(I))*(.25*V(I-1,JM,K)+.5*V(I,JM,K)
&           +.25*V(I+1,JM,K))
END DO

```

The corresponding coding for the tangential baroclinic velocity at the southern and northern boundaries are as below.

```

ck ** South
DO I = 2, NBS
  AIUS(I) = 2.*DTI/(DY(I-1,1)+DY(I,1)+DY(I-1,2)+DY(I,2))
  VMID    =( V(I,2,K)+V(I-1,2,K) )/2.
  UF(I,1,K)=U(I,1,K) - AIUS(I) *
&           ( (VMID+ABS(VMID))*(U(I,1,K)-0.E0      )
&           +(VMID-ABS(VMID))*(U(I,2,K)-U(I,1,K)) )
END DO

ck ** North
DO I = 2, NBN
  VMID=.5E0*(V(I,JM,K)+V(I-1,JM,K))
  UF(I,JM,K)=U(I,JM,K) - AIUN(I) *
&           ( (VMID+ABS(VMID))*(U(I,JM,K)-U(I,JMM1,K))
&           +(VMID-ABS(VMID))*(0.E0      -U(I,JM,K)) )
END DO

```

Note the 2 interior points closest to the domain boundary are used for the normal BCs at the southern and western boundaries because of the way the C grid is defined for the POM (see Appendix B).

As in the case of the barotropic velocities the values of v at the first row of the southern boundary and values of u at the first column of the western boundary are assigned the values of their adjacent grid points for numerical safety.

```

DO I = 1, IM
  VF(I,1,K) = VF(I,2,K)
END DO
DO J = 1, JM
  UF(1,J,K) = UF(2,J,K)
END DO

```

There is no suitable information on vertical profiles of river inflows to use for the baroclinic velocities. The values of u at the river locations are therefore assigned to those of the adjacent interior grid point where they are computed by the POM. This is coded as below.

```

DO I = 1, NRV
  UF(IRV(I),JRV(I),K) = UF(IRV(I)-1,JRV(I),K)
END DO

```

5. TEMPERATURE AND SALINITY

5.1 Water Column

An advective BC is applied along the open boundary for the temperature T and salinity S throughout the water column. It is identical in form to that used in (12) for the tangential baroclinic velocity

$$\frac{\partial T}{\partial t} + v_n \frac{\partial T}{\partial n} = 0 \quad (14)$$

$$\frac{\partial S}{\partial t} + v_n \frac{\partial S}{\partial n} = 0 \quad (15)$$

The normal gradients of the temperature and salinity are given here by

$$\frac{\partial T}{\partial n} = \begin{cases} (T - T_0)/\Delta x_n & v_n \geq 0 \\ (T_i - T)/\Delta x_n & v_n < 0 \end{cases} \quad (16)$$

$$\frac{\partial S}{\partial n} = \begin{cases} (S - S_0)/\Delta x_n & v_n \geq 0 \\ (S_i - S)/\Delta x_n & v_n < 0 \end{cases} \quad (17)$$

where T_0 and S_0 here are the climatological temperature and salinity at the boundary, and T_i and S_i the temperature and salinity at one grid point inside the open boundary, respectively. The advective BC is applied to T and S at the southern, northern and western boundaries.

The FORTRAN implementation of these BCs at the southern boundary is as below:

```

DO I = 2, NBS
    ATSS(i) = DTI/(DY(I,1)+DY(I,2))
    UF(I,1,K) = T(I,1,K) - ATSS(I) *
    &           ( ( V(I,2,K)+ABS(V(I,2,K)) )*(T(I,1,K)-TBS(I,K))
    &           + ( V(I,2,K)-ABS(V(I,2,K)) )*(T(I,2,K)-T(I,1,K)) )
    VF(I,1,K) = S(I,1,K) - ATSS(I) *
    &           ( ( V(I,2,K)+ABS(V(I,2,K)) )*(S(I,1,K)-SBS(I,K))
    &           + ( V(I,2,K)-ABS(V(I,2,K)) )*(S(I,2,K)-S(I,1,K)) )
END DO

```

Note that the temporary variables UF and VF are used to contain T and S on entry into the BCOND subroutine.

As in the case of the elevation, the SW and NW corners of the model grid are averaged with their boundary neighbours.

```

UF(1, 1,K) = (UF(2,1,K)+UF(1,2,K))/2.
VF(1, 1,K) = (VF(2,1,K)+VF(1,2,K))/2.
UF(1,JM,K) = (UF(2,JM,K)+UF(1,JMM1,K))/2.
VF(1,JM,K) = (VF(2,JM,K)+VF(1,JMM1,K))/2.

```

The temperature and salinity at the mouths of rivers is specified in the PWC POM. For temperature the value is specified to be that given by river data, with a blending of the ocean and river temperatures being used during the ramp up phase of the model. For salinity the river is considered to be pure fresh water. This BC is coded as

```

DO I = 1, NRV
  DO K = 1, KBM1
    UF(IRV(I),JRV(I),K) = UF(IRV(I),JRV(I),K)
    &           + (TRV(K,I) - UF(IRV(I),JRV(I),K))
    &           * RAMP
    VF(IRV(I),JRV(I),K) = VF(IRV(I),JRV(I),K)*(1.-RAMP)
  END DO
END DO

```

where TRV is the temperature of the river.

5.2 Sea Surface Temperature and Salinity

The surface temperature and salinity used in the POM are the values of the profiles in the previous subsection with a relaxation to externally supplied fields of SST and SSS. In the original implementation a spatially filtered SST field (SST_f) is first produced from the surface values of the profiles using a gaussian weight distribution having an attenuation length of 5 grid points. This was done because the original MCSST used was created from 5 day composites of the satellite images and was rather noisy. The PWC POM of CoBALT uses daily MCSST from the Modular Ocean Data Assimilation System which do not have such problems. No spatial filtering is therefore applied in the PWC POM when using daily MCSST ($SST_f = MCSST$). No spatial filtering is applied to the SSS values of the profiles. The sea surface temperature and salinity fields (SST_0 and SSS_0) are then assimilated into the model via a relaxation method

$$\frac{\partial SST}{\partial t} = \lambda_T (SST_0 - SST_f) \quad (18)$$

$$\frac{\partial SSS}{\partial t} = \lambda_S (SSS_0 - SSS). \quad (19)$$

In the original implementation of the PWC POM time scales of $\lambda_T^{-1} = 5$ d and $\lambda_S^{-1} = 30$ d were used. When the PWC POM is forced with daily MCSST the value $\lambda_T^{-1} = 1$ d is used instead. (Note that a two timesteps leapfrog scheme is used in the finite difference forms of the POM.) The SST_0 and SSS_0 fields are linearly interpolated to the model time step. Note that the relaxation is applied independent of the model time step. The PWC POM presently assimilates daily MCSST and monthly Levitus SSS.

5.3 Interior Relaxation to Climatology

To prevent model drift away from the seasonal cycle the subsurface temperature and salinity are relaxed towards the Levitus monthly mean temperature (\bar{T}) and salinity (\bar{S})

$$\frac{\partial T}{\partial t} = \lambda_{ST} (\bar{T} - T) \quad (20)$$

$$\frac{\partial S}{\partial t} = \lambda_{ST} (\bar{S} - S). \quad (21)$$

The time scale for the relaxation is given by

$$\lambda_{ST} = \frac{1}{t_{ST}} \left(1 - \exp \left(\frac{z}{z_0} \right) \right), \quad (22)$$

where $t_{ST} = 250$ d, $z_0 = 500$ m, and $z < 0$. The 250 day time scale was found to be adequate for having the interior temperature and salinity follow the seasonal cycle. The time scale is exponentially weighted with depth to ensure T and S at the surface only relaxes to the externally supplied SST and SSS at the surface. A check is subsequently made on the resulting values of salinity and values of $S < 0$ are set to zero.

6. VERTICAL VELOCITY

A zero gradient BC in the horizontal is applied to the vertical velocity w along all 3 open boundaries of the model domain

$$\nabla_H w = \mathbf{0} \quad (23)$$

where ∇_H is the horizontal gradient. Vertical velocities at river locations are assigned to those of the adjacent interior grid point where they are computed by the POM.

These BCs are coded within the model as below.

```

DO 245 K=1,KBM1
      DO 246 J=1,JM
246      W(1,J,K)=W(2,J,K)
      DO 247 I=2,IM
            W(I,JM,K)=W(I,JMM1,K)
247      W(I,1,K)=W(I,2,K)
245 CONTINUE
C
C      Rivers
      DO I = 1, NRV
        DO K = 1, KBM1
          W(IRV(I),JRV(I),K) = W(IRV(I)-1,JRV(I),K)
        END DO
      END DO
    
```

7. TURBULENT KINETIC ENERGY AND LENGTH SCALE

A clamped BC is applied to twice the turbulent kinetic energy q^2 and its product with the turbulence length scale l along the open boundaries. Values of q^2 and $q^2 l$ at rivers are assigned those of the adjacent interior grid point.

The FORTRAN coding of these BCs is given below.

```

DO 300 K=1,KB
      DO 295 J=1,JM
            UF(IM,J,K)=1.E-10
295      VF(IM,J,K)=1.E-10
      DO 296 I=1,IM
            UF(I,JM,K)=1.E-10
            VF(I,JM,K)=1.E-10
            UF(I,1,K)=1.E-10
    
```

```
296      VF(I,1,K)=1.E-10
C
C      Rivers
DO I = 1, NRV
    UF(IRV(I),JRV(I),K) = UF(IRV(I)-1,JRV(I),K)
    VF(IRV(I),JRV(I),K) = VF(IRV(I)-1,JRV(I),K)
END DO
300  CONTINUE
```

Note that the temporary variables UF and VF are used to hold q^2 and $q^2 l$ on entry into the BCOND subroutine.

8. SUMMARY

We have presented here the explicit details of how externally supplied forcing has been implemented for one way coupling along the open and coastal boundaries of the PWC POM. These forcing fields derive from a variety of sources that include a coarser resolution OGCM, satellite derived fields, and river inflows. A variety of formulations are implemented in the PWC POM that include advectional, clamped, Flather, and radiational BCs, as is appropriate for each variable. Most of these BCs have been implemented in some form in the Navy Coordinate Ocean Model (NCOM) and it is hoped that this report will be of value to NCOM users as well.

9. ACKNOWLEDGEMENTS

The authors would like to thank Dong-Shan Ko for developing the first version of the PWC POM and for providing the figure and material for Appendix B. This contribution was funded through the Naval Research Laboratory (NRL) 6.1 Coupled Biophysical-Dynamics Across the Littoral Transition (CoBALT) Accelerated Research Initiative under program element 61153N sponsored by the Office of Naval Research. This manuscript is NRL contribution number JA/7330-00-8245.

10. REFERENCES

Allard, R., P. Farrar, D. Mark, J. Martin, B. Shapiro, S. Lowe and D.S. Ko, Creating a 4-D integrated scenario near Camp Pendleton, CA, in *Conference on Coastal Oceanic and Atmospheric Prediction*, American Meteorological Society, 1996.

Clancy, R. M., P. W. deWitt, P. W. May and D. S. Ko, Implementation of a coastal ocean circulation model for the West Coast of the United States, in *Conference on Coastal Oceanic and Atmospheric Prediction*, American Meteorological Society, 1996.

Flather, R. A., A tidal model of the northwest European continental shelf, *Mern. Soc. R. Sci. Liege*, Ser. 6, **10**, 141–164, 1976.

Haidvogel, D. B., J. Blanton, J. C. Kindle and D. R. Lynch, Coastal Ocean Modeling: Processes, and Real-time Systems. *Oceanography*, **13**, No. 1, 35–46, 2000.

Kindle, J. C., S. C. Cayula, J. E. Metzger and O. M. Smedstad, A nested Model of the California Current System: Sensitivity to Boundary Information, *EOS Transactions*, American Geophysical Union, **80**, No. 49, December 7 ,1999.

Ko, D. S., Coupling POM to the NRL Global Layer Model: Application to the West Coast, in *Proceedings from the Princeton Ocean Model (POM) Users Meeting*, Princeton Univ., 1996a.

Ko, D. S., R. C. Rhodes, R.A. Allard, E.J. Metzger, H.E. Hurlburt, R.M. Clancy and P.W. deWitt, A coupled ocean model for the U.S. West Coast coastal prediction, in *IEPB-6 sponsored Ocean Modeling Workshop*, Monterey, CA, 1996b.

Ko, D. S., R. A. Allard, E. J. Metzger and R. C. Rhodes, A coupled West Coast model with seasonal forcing, *EOS, (suppl.)*, **76**, American Geophysical Union, 1996c.

Ko, D. S. R. Preller, M. Carnes: Nowcast/Forecast Experiment applying a Pacific West Coast Model, in *6th International Conference on Estuarine and Coastal Modeling*, New Orleans, Louisiana, November, 1999a.

Ko, D. S. R. Preller, M. Carnes, C. Barron and P. Posey: An Experimental Real-Time North Pacific Ocean Nowcast/Forecast System, in *Proceedings from the Sigma Coordinate Ocean Model Users Meeting 99*, Bar Harbor, Maine, 1999b.

Levitus, S., R. Burgett, and T. P. Boyer, *World Ocean Atlas 1994*, vol. 3, *Salinity, NOAA Atlas NESDIS 3*, 99 pp., U.S. Govt. Print. Off., Washington, D.C., 1994.

Levitus, S., and T. P. Boyer, *World Ocean Atlas 1994*, 4, *Temperature, NOAA Atlas NESDIS 4*, 117 pp., U.S. Govt. Print. Off., Washington, D.C., 1994.

Metzger, E. J., R. C. Rhodes, D. S. Ko and H.E. Hurlburt, *Validation Test Report for OCEANS 1.0: The 1/4 degree Global, Reduced Gravity NRL Layered Ocean Model, NRL Tech. Rep., NRL/FR/7323-97-9673*, Naval Research Laboratory, Stennis Space Center, 31 pp., 1998.

Righi, D. D., P. T. Strub and J. C. Kindle, Validation of a California Current model through comparison with altimeter and drifter circulation statistics, in *Proceedings of the 3RD Conference on Coastal Atmospheric and Oceanic Prediction Processes*, 2-5 Nov., 1999, New Orleans, 1999.

Shulman, I., C. R. Wu, J. K. Lewis, J. D. Paduan, L. K. Rosenfeld, S. R. Ramp, M. S. Cook, J. C. Kindle, and D. S. Ko, Development of the High Resolution, data Assimilating Numerical Model of the Monterey Bay. *Estuarine and Coastal Modeling*, in press, 1999a.

Shulman, I., Data Assimilation and Coupling the Global NRL Model with the Regional Princeton Ocean Model, Office of Naval Research, *Fiscal Year 1998 Annual Reports*, 1999b.

Appendix A

FORTRAN SYMBOLS

The FORTRAN variable names used in the POM that are cited within this report are listed below for ease of reference. The variables are listed as given in the POM Users Guide [Mellor, 1998] and the reader is referred to the latter for a more complete list. The variables are given followed by their corresponding analytical symbols in parentheses and a brief explanation. Not explicitly tabulated are the suffixes B, blank and F which are appended to many of the variables to denote the previous ($n - 1$), latest (n), and forward ($n + 1$) time levels, respectively.

Indices

I,J (i,j)	horizontal grid indices
IM,JM	outer limits of I and J
IMM1,JMM1	IM-1 and JM-1
IRV,JRV	grid indices of river locations
K (k)	vertical grid index; K=1 at the top and K=KB at the bottom
NBS, NBN, NBW	the total number of ocean grid points along the S, N, and W boundaries

Constants

DTE (Δt_E)	external mode time step (s)
DTI (Δt_I)	internal mode time step (s)
GRAV (g)	acceleration due to gravity ($m s^{-2}$)
RAMP (λ)	amplitude during spin up from rest ($0 < \lambda < 1$)

One-dimensional Arrays

ELGS (η_0)	externally elevation at open boundary (m)
UAGS, VAGS (U_0, V_0)	externally supplied transport at open boundary ($m s^{-1}$)

Two-dimensional Arrays

DX (Δx)	longitude grid spacing (m)
DY (Δy)	latitude grid spacing (m)
EL (η)	surface elevation as used in the external mode (m)
H (H)	the bottom depth (m)
TRV	the temperature of the river ($^{\circ}C$)
UA, VA (U, V)	vertical mean of u, v ($m s^{-1}$)

Three-dimensional Arrays

U, V (u, v)	horizontal velocities ($m s^{-1}$)
W (ω)	sigma coordinate vertical velocity ($m s^{-1}$)

A.1 REFERENCES

Mellor, G. L., *Users guide of a three-dimensional primitive equation, numerical ocean model*, Program in Atmospheric and Oceanic Sciences, Princeton University, Princeton, N.J., (available at <http://www.aos.princeton.edu/WWWPUBLIC/htdocs.pom>), 1998.

Appendix B

PRINCETON OCEAN MODEL C-GRID

The POM uses the C-grid shown in Figure B1. The locations of the velocity components are indicated by U and V, and the mass point locations by the elevation E. The temperature and salinity are at the mass points and are therefore coincident with E. The diagram shown is for the case where velocities are the primary boundary conditions together with temperature, salinity, and elevation. The elevation itself is rather unimportant and is substituted from an adjacent interior point.

The symbols appearing within the dotted box correspond to the interior (non-boundary) grid points. In general only those variables in the interior are computed and variables at the open boundary have to be specified. All interpolations are centered in space except those at lateral open boundaries where an upstream scheme is usually used. The BC symbols indicate the boundaries for the various variables on the C-grid. The rows/columns indicated by NU are not used in the model calculations as are variables marked by an asterix (*). Note that adjacent interior values may be filled in at these latter points for more attractive output of the fields.

Fig. B1 – The C-grid Used by the PWC POM.

